
Subject: Re: delete a tagname from a structure?
Posted by Fabzi on Tue, 24 Sep 2013 08:44:34 GMT
View Forum Message <> Reply to Message

Hi

On 09/23/2013 11:55 PM, spluque@gmail.com wrote:
>  Searching for a way to remove a field from a structure,

I found this in my lib, I don't remember who wrote the code in the first 
place but it's not that complex:

;+
; :Description:
;    Removes a tag from a structure.
;
; :Params:
;    struct: in, required
;            the structure
;    tagname: in, required, type=string
;             the tag name
;
;-
pro w_remove_tag, struct, tagname

   ; Set Up environnement
   COMPILE_OPT idl2

   if n_params() ne 2 then MESSAGE, 'N_PARAMS=2'

   searchtag=strupcase(tagname)
   tagnames=tag_names(struct)
   a=[-1]
   if n_elements(tagname) eq 1 then a=[a,where(tagnames $
     ne searchtag)] else for i=0,n_elements(tagnames)-1 do $
     if (where(searchtag eq tagnames[i]))[0] eq -1 then a=[a,i]
   if n_elements(a) eq 1 then return
   if a[1] eq -1 then return
   newstruct=create_struct(tagnames[a[1]],struct.(a[1]))
   if n_elements(a) gt 2 then for i=2,n_elements(a)-1 $
     do  newstruct=create_struct(newstruct,tagnames[a[i]],struct.(a[i ]))
   struct=newstruct

end

